

STATISTICAL BULLETIN

Reliability & Variation Research

LEONARD G. JOHNSON
EDITOR

DETROIT RESEARCH INSTITUTE
P.O. Box 36504 • Grosse Pointe, MI 48236 • (313) 886-8435

WANG H. YEE
DIRECTOR

Copyright © 1971 - 1998

Volume 28
Bulletin 4

August, 1998
Page 1

A COMPUTER PROGRAM FOR DETERMINING THE ORDER STATISTIC NUMBER CORRESPONDING TO THE PRODUCT OF TWO VALUES TAKEN FROM TWO SAMPLES, ONE FROM EACH SAMPLE, IN A TOTAL SAMPLE WHOSE SIZE IS EQUAL TO THE PRODUCT OF THE TWO INDIVIDUAL SAMPLE SIZES FROM WHICH THE TWO MULTIPLIED VALUES CAME

INTRODUCTION

To define the nature of the problem which we are considering in this bulletin, let us look at the two lines shown in **Figure 1**.




Figure 1

Line #1 represents a sample of Size N_1 , with N_1 order statistics, while Line #2 represents a sample of size N_2 , with N_2 order statistics.

Now, if we multiply each item in Sample #1 by each item in Sample #2, we get a total of N_1N_2 products. In this list of N_1N_2 products we want to know the order number of the product obtained by multiplying order statistic J_1 in the first sample of size N_1 by order statistic J_2 in the second sample of size N_2 .

THE MATHEMATICAL SOLUTION TO THE PROBLEM

In our July 1998 Statistical Bulletin (Volume 28, Bulletin 3) we derived the formula for the desired order statistic # x in sample size N_1N_2 . This formula is given in (1) below:

$$x = \frac{J_1 J_2 (N_1 N_2 + 1)}{J_1 J_2 + (N_1 + 1 - J_1)(N_2 + 1 - J_2)} \quad (1)$$

So, in order to write a simple basic computer program to determine the value of x in (1), all we need to do is given the input values

N_1 = Sample Size No. 1

J_1 = Order Statistic No. in Sample #1

N_2 = Sample Size No. 2

J_2 = Order Statistic No. in Sample #2

With these input values as starters, we design a computer program in GW-BASIC as listed on Page 3.

DRI STATISTICAL BULLETIN

Volume 28

Bulletin 4

August, 1998

Page 3

THE GW-BASIC COMPUTER PROGRAM

```
5 COLOR 15,1:CLS
10 PRINT "MULTIPLIED ORDER STATISTICS PROGRAM - (MOSTATS)"
20 LPRINT "MULTIPLIED ORDER STATISTICS PROGRAM - (MOSTATS)"
30 INPUT "FIRST SAMPLE SIZE =";N1
40 INPUT "ORDER STATISTIC NO. IN FIRST SAMPLE =";J1
50 INPUT "SECOND SAMPLE SIZE =";N2
60 INPUT "ORDER STATISTIC NO. IN SECOND SAMPLE =";J2
70 LPRINT "FIRST SAMPLE SIZE =";N1
80 LPRINT "ORDER STATISTIC NO. IN FIRST SAMPLE =";J1
90 LPRINT "SECOND SAMPLE SIZE =";N2
100 LPRINT "ORDER STATISTIC NO. IN SECOND SAMPLE =";J2
110 C1=(N1+1-J1)/(N1+1)
120 PRINT "CONFIDENCE LEVEL FOR THE ORDER STATISTIC IN FIRST SAMPLE =";C1
130 LPRINT "CONFIDENCE LEVEL FOR THE ORDER STATISTIC IN FIRST SAMPLE =";C1
140 C2=(N2+1-J2)/(N2+1)
150 PRINT "CONFIDENCE LEVEL FOR THE ORDER STATISTIC IN SECOND SAMPLE =";C2
160 LPRINT "CONFIDENCE LEVEL FOR THE ORDER STATISTIC IN SECOND SAMPLE =";C2
170 X=(J1*J2*(N1*N2+1))/(J1*J2+(N1+1-J1)*(N2+1-J2))
180 N=N1*N2
190 PRINT "TOTAL MULTIPLIED SAMPLE SIZE =";N
200 LPRINT "TOTAL MULTIPLIED SAMPLE SIZE =";N
210 PRINT "RESULTANT ORDER STATISTIC NO. IN TOTAL MULTIPLIED SAMPLE SIZE =";X
220 LPRINT "RESULTANT ORDER STATISTIC NO. IN TOTAL MULTIPLIED SAMPLE SIZE =";X
230 C=(N+1-X)/(N+1)
240 PRINT "RESULTANT CONFIDENCE LEVEL IN TOTAL MULTIPLIED SAMPLE SIZE =";C
250 LPRINT "RESULTANT CONFIDENCE LEVEL IN TOTAL MULTIPLIED SAMPLE SIZE =";C
260 PRINT
280 LPRINT
300 INPUT "ANY MORE CALCULATIONS DESIRED - Y/N";Y$
310 IF Y$="Y" OR Y$="Y" THEN 10
320 PRINT "END OF CALCULATIONS"
330 LPRINT "END OF CALCULATIONS"
350 END
```

DRI STATISTICAL BULLETIN

Volume 28
Bulletin 4

August, 1998
Page 4

AN ILLUSTRATED EXAMPLE

QUESTION: Suppose we multiplied order statistic # 10 in a sample of 40 by order statistic # 15 in a sample of 70.

What order statistic number does this product represent in the multiplied sample size, i.e., 2800?

The computer printout for this example is listed below:

```
MULTIPLIED ORDER STATISTICS PROGRAM - (MOSTATS)
FIRST SAMPLE SIZE = 40
ORDER STATISTIC NO. IN FIRST SAMPLE = 10
SECOND SAMPLE SIZE = 70
ORDER STATISTIC NO. IN SECOND SAMPLE = 15
CONFIDENCE LEVEL FOR THE ORDER STATISTIC IN FIRST SAMPLE = .7560976
CONFIDENCE LEVEL FOR THE ORDER STATISTIC IN SECOND SAMPLE = .7887325
TOTAL MULTIPLIED SAMPLE SIZE = 2800
RESULTANT ORDER STATISTIC NO. IN TOTAL MULTIPLIED SAMPLE SIZE = 222.7731
RESULTANT CONFIDENCE LEVEL IN TOTAL MULTIPLIED SAMPLE SIZE = .9204667

END OF CALCULATIONS
```

Note that the resultant confidence .9204667 is the same resultant which would be obtained by multiplication of odds from the two confidence levels in the two samples.

Thus, First Confidence = $C_1 = .7560976$
This is $ODDS_1 = .7560976 / .2439024 = 3.1$

Also, Second Confidence = $C_2 = .7887325$
This is $ODDS_2 = .7887325 / .2112675 = 3.73333$

Then, RESULTANT ODDS = $ODDS_1 \times ODDS_2 = 11.5733$

Therefore, RESULTANT CONFIDENCE = $11.5733 / 12.5733 = .920466$

This agrees with the last line in the computer printout.

CONCLUSION

We have succeeded in designing a comprehensive computer program which combines and applies all of the mathematical principles involved in the combining of two independent samples by the multiplication of any two values (order statistics), one from each sample, so as to become a resultant order statistic in the resultant total sample size, which is the product of the two individual sample sizes.

This is another rigorous verification of the fact that the resultant odds for the resultant reliability obtained by multiplying two independent reliabilities is equal to the product of the two separate odds given for the two independent reliabilities in series.